HP 42S Topography



D'Zign

TECHNICAL ASSISTANCE

The program material, instructions and procedures contained in this book assume that the user has a working knowledge of both surveying *and* the general operation of the HP-42S calculator.

Technical assistance is limited to verification of the results shown in the various examples used in this book.

If you have any questions or suggestions regarding this book, or other **D'Zign** publications, please feel free to call us. The number is (209) 297-8025, and someone is available to answer technical questions between the hours of ***8:00 A.M. and Noon***, (Pacific Time Zone), ***Monday through Thursday***.

Before calling for help, take a look through *"The Most Commonly Asked Questions"*, on the inside of the back cover, and the comment about debugging on page 20.

WARNING

NOTICE

HP42S Topography

Library of Congress Catalog Card No. 89-81215
ISBN 0-944889-09-3

Manufactured in the United States of America

published by D'Zign Land Survey & Development
Glendale, California, U.S.A. 1990

ISBN 0-944889-09-3

The programs included in this booklet are designed to take full advantage of the power of the Hewlett-Packard HP42S calculator. Programming this calculator is really simple, but a bit confusing at first. We will try to walk you through some of the 'harder to find' steps as we proceed.

the operations index

To find a function for the first time, HP has provided an "Operations Index" on pages 310 through 335 of the instruction manual, which tells you exactly what keystrokes to use to type in the function you want.

Even better, this index gives you the page number that you can refer to if you want to know more about the function you are using. If, while typing in a program, you aren't sure how to input a particular function, simply refer to the Operations Index.

Another handy tool is the function catalog. When you stroke the shifted + key, a menu appears in the lower portion of the display. The leftmost key will take you into the function catalog, which contains ALL of the functions. Scroll up or down through the list until you come to the function you want, stroke the corresponding key, and the function is entered as a program step.

the programs

If this is your first try at programming the 42, we recommend that you read Chapter 8 of the manual before beginning.

The programs included in this booklet have been separated into accessable sub-programs to allow them to be used with other programs at a later date. For instance, if you are using the **HP42S Alignment & Offsets** booklet, you will find that a number of the subroutines you need for this program are already in the calculator.

A number of the subroutines will already have been input if you are using programs from the book, "**HP42S Surveying Solutions**". If a program or subroutine has the same NAME as one you already have (from any **D'Zign** publication) it is the same as the one in this book.

subroutines

Because of the way the calculator works, we will start by input of some subroutines.

Once the subroutine has been input, its name appears in the menu when you stroke XEQ, and all you have to do to add it as a step in the program you are typing in is stroke GTO or XEQ followed by keystroking the key corresponding to the subroutine to input the program step GTO XXX or XEQ XXX.

```
01▸LBL "YN"
02 "YES"
03 KEY 1 GTO 01
04 "NO"
05 KEY 2 GTO 02
06 MENU
07 STOP
08▸LBL 01
09 SF 10
10 GTO 03
11▸LBL 02
12 CF 10
13▸LBL 03
14 CLMENU
15 EXITALL
16 RTN
```

getting started

Begin by stroking the shift key, then the **XEQ** key. The display will show a menu which will be blank (if you haven't yet input any programs) except for **.END.** on the left. The keys just below each of the menu portions will correspond to the menu instruction above it. Stroke the key just below the **.END.** in the display.

```
00▸{ 0-Byte Prgm }
01 .END.
```

Next, go into **program mode** by stroking the shifted R/S key, and you should have a display similar to the one shown to the left. Begin typing in the program "**YN**" from the listing above.

quick tip

Program steps 03 and 05 use a function which stores the prompt to the menu, and at the same time assigns the key.

To access the function, stroke ⬜ PGM.FCN ▲ KEY

You'll receive a prompt, **KEY _ .** Stroke the key number (we'll use 1 as the example), and you will get a prompt, **KEY 1 GTO _ _.**

In this case (step 03) answer 01 to complete the program step.

next subroutine

This one has 2 steps you'll want to review before you begin input:

02 ΣREG 00	To access this function, go to the "stat" menu (shifted divide key), and scroll down once. It's the second key from the right, and when you stroke it you will be prompted for the 00 to complete the program step.
05 CLΣ	This one is the leftmost key when you bring up the menu by stroking [ ] [←].

To begin input, stroke the **shifted** [XEQ] key, then the key that corresponds to the menu listing **"YN"**, the program just input.

Scroll upward once with the [▲] key to put the pointer at step 00, and begin typing in the program steps shown to the right.

When you've finished stroke [EXIT] to leave program mode.

```
01▶LBL "CL"
02 ΣREG 00
03 CLΣ
04 ΣREG 11
05 CLΣ
06 0
07 STO 24
08 RTN
```

The step, **CF IND ST X**, (in the next group) is input through the FLAGS menu. To get to "IND", stroke the [·] key, then stroke it again to bring up the menu containing "ST X".

Go back into program mode ([ ] [R/S]). The program pointer should still be at step **08 RTN**. Type in the additional steps shown on the next page.

```
09▶LBL "FC0"
10 0.013
11▶LBL "FN0"
12 CF IND ST X
13 ISG ST X
14 GTO "FN0"
15 GTO IND ST X
16▶LBL 14
17 CF 19
18 CF 20
19 81.088
20 GTO "FN0"
21▶LBL 89
22 90.098
23 GTO "FN0"
24▶LBL 99
25 CLST
26 FS? 55
27 SF 08
```

We're going to add one more step, 28 END. Input this step by stroking XEQ ENTER, type in END, and then stroke ENTER again.

Input of the "**END**" step has separated this program from the program "**YN**". This general method of input will be used for almost all of the programs, starting at the 'top' of one program and then separating the two programs with an END as the last step of the new program. Using this method, we can put the programs in the menu where we want them.

cleaning house

Next, we want to do some editing that will make life easier later. Go to "**YN**", and then scroll upward to put the pointer at step 16, RTN. Type in a new label, "**FILE**".

Scroll up again to the RTN, delete it, and replace it with END the same way you just did after "**FC0**". This should leave the new label as the first step in the display, having separated this new label from the rest of the programming.

```
01▶LBL "FILE"
02 CF 21
03 "INPUT "
04 ASTO 28
05 ⊢"FILE NAME"
06 AVIEW
07 CLA
08 AON
09 STOP
10 AOFF
11 ASTO 26
12 ASHF
13 ASTO 27
14 CLA
15 CLX
16 ARCL 28
17 ⊢"DATE"
18 AVIEW
19 SF 21
20 STOP
21 STO 28
22 .END.
```

Finish typing in the program shown to the left (note that the **.END.** is already there, at the bottom).

Another new symbol that you'll be using a lot is the ⊢ symbol. It adds to what is already in the **alpha** register without overwriting it. To input the symbol, stroke ENTER to enter **alpha** mode, and then stroke ENTER again.

After you type in step 21, STO 28, if you scroll down one you should see the **.END.** as step 22. We will leave it there, in place of a regular END command; doing so will keep it out of the menu display from now on.

Another one to look at; the symbol, ↳ ,is "line feed".

```
irst Shot #?↳"
OMPT
ast Shot #?↳"
OMPT
```

We use it to control the display. You can input it by stroking

▢ ALPHA ▼ ▼

Go to "**FILE**", enter **program mode**, and then scroll upwards with the ▲ key until the pointer is at 21 STO 28. Now type in LBL "TOUT", scroll back up to 21 STO 28 and add an END. Stroke EXIT ▢ XEQ TOUT ▢ R/S , and then type in the rest of the program.

```
01▶LBL "TOUT"
02 XEQ "TO"
03 "First Shot #?↳"
04 PROMPT
05 "Last Shot #?↳"
06 PROMPT
07 1000
08 ÷
09 +
10 STO 13
11▶LBL a
12 SF 21
13 ADV
14 CLA
15 "↳Shot #"
16 FIX 00
17 CF 29
18 ARCL 13
19 SF 29
20 FIX 04
21 AVIEW
22 RCL 13
23 RCLX 01
24 RCL+ 02
25 STO 24
26 FS? 83
27 XEQ 05
28 FS? 86
29 XEQ 07
30 FS? 87
31 XEQ 06
32 ISG 13
33 GTO a
34 CLST
35 CLA
36 FIX 04
37 "END OF FILE↳"
38 ├"  "
39 ARCL 26
40 ARCL 27
41 ADV
42 AVIEW
43 RTN
44▶LBL 05
45 CLA
46 "H<= "
47 RCL IND 24
48 XEQ "DMS"
49 DSE 24
50 STO ST X
51 ├"↳Z<= "
52 RCL IND 24
53 XEQ "DMS"
54 AVIEW
55 FIX 02
56 DSE 24
57 STO ST X
58 "S.D.= "
59 ARCL IND 24
60 ├"↳ROD= "
61 DSE 24
62 STO ST X
63 ARCL IND 24
64 AVIEW
65 DSE 24
66 STO ST X
67 RTN
68▶LBL 06
69 ARCL IND 24
70 DSE 24
71 STO ST X
72 ARCL IND 24
73 DSE 24
74 STO ST X
75 ARCL IND 24
76 ├"↳"
77 AVIEW
78 RTN
79▶LBL 07
80 CLA
81 FIX 04
82 ADV
83 FC? 20
84 XEQ 01
85 FS? 20
86 XEQ 02
87 AVIEW
88 CLA
89 DSE 24
90 STO ST X
91 FIX 02
92 "  EL= "
93 ARCL IND 24
94 ├"↳"
95 AVIEW
96 DSE 24
97 STO ST X
98 CLA
99 RTN
100▶LBL 01
101 "  N = "
102 ARCL IND 24
103 ├"↳"
104 DSE 24
105 STO ST X
106 ├"  E = "
107 ARCL IND 24
108 RTN
109▶LBL 02
110 RCL IND 24
111 CLA
112 "Sta. "
113 XEQ "STA"
114 DSE 24
115 STO ST X
116 RCL IND 24
117 FIX 02
118 RND
119 X<0?
120 SF 99
121 FS? 99
122 +/-
123 ├"↳O/S = "
124 ARCL ST X
125 FS? 99
126 ├" Left"
127 FC?C 99
128 ├" Right"
129 RTN
130▶LBL 03
131 CF 20
132 SF 19
133 RTN
134▶LBL 04
135 CF 19
136 SF 20
```

If you scroll down with ▼ you'll see that "**TOUT**" now has the permanent **.END.**.

Continuing with input, the program shown to the right is a subroutine that lets you decide what type of data will be stored. This is another that may be put in on top of "**YN**", and it has an END to separate it when you are finished with the input.

During use of the program, this subroutine will bring up a menu bar, , as one of the prompts, allowing you to select which kinds of data you want to store.

Stroke the key beneath each of the types you want, and then stroke R/S to continue. This program automatically allocates the correct number of registers to hold data for each shot.

Selecting all three (raw, finished and descriptor) will use ten storage registers for each shot. The raw data stored will be the horizontal and vertical angles, the slope distance and the rod reading.

Finished data can either be as 3-dimensional coordinates or in the form of station, offset and elevation. This option is offered by this subroutine also, using the menu .

Select any (or all) of the options by stroking the keys associated with the menu selections you want and then stroke R/S to continue with the program.

Don't forget to proof-read as you go.

```
▶LBL "TY"
CLA
CF 20
├" TYPE"
ASTO 07
CLMENU
"COORD"
KEY 1 GTO 03
"S-O/S"
KEY 3 GTO 02
MENU
"SELECT TOPO"
ASTO 08
ARCL 07
CF 21
PROMPT
▶LBL 02
SF 20
▶LBL 03
CLMENU
CLA
"RAW"
KEY 1 GTO 01
"FIN'D"
KEY 3 GTO 00
"DESC."
KEY 5 GTO 05
MENU
CLA
ARCL 08
├" DATA"
ARCL 07
AVIEW
STOP
▶LBL 00
SF 86
2
GTO 04
▶LBL 01
SF 83
1
GTO 04
▶LBL 05
SF 87
4
▶LBL 04
STO+ 24
CLX
AVIEW
STOP
RCL 24
5
X<>Y
X=Y?
85
STO 25
STO 24
SF 21
SF IND 24
CLST
CLMENU
END
```

Topography (the program) is, as you can see, a combination of shorter programs and subroutines. By keeping each program as short as possible (using separated subroutines) the 'parent' program runs faster.

```
01▶LBL "STA"
02 CF 29
03 FIX 00
04 STO 21
05 1E2
06 ÷
07 ENTER
08 IP
09 ARCL ST X
10 -
11 ⊢"+"
12 FIX 03
13 1E2
14 x
15 10
16 X>Y?
17 ⊢"0"
18 ARCL ST Y
19 RCL 21
20 SF 29
21 FIX 04
22 RTN
```

You may also add programs of your own that use these same subroutines to do a specific task just by having your program execute an already existing program. This is one that you may want to use as a subroutine for your own programs, later.

"**STA**" changes the number in the x-register into the form we use for stationing. It can go in right on top of "**FILE**".

Should you decide to use "**STA**" for a different program, notice that it does not have an AVIEW at the end, nor does it clear the **alpha** register before execution. Those two functions need to be added to your parent program when you use this subroutine.

We haven't added an END to the program either, because we'll be adding more to it later.

Steps 05 and 13 are input using the **E** key on the keyboard. It isn't necessary to input the 1, just stroke **E**, followed by the number (in this case, 2). Nothing will happen until you stroke the key for the next program step, usually **×** or **÷**.

The next program changes the number in the x-register to the ° ' " form for output of angles. It doesn't have an AVIEW either, but the clearing of the **alpha** register is optional. If you want it cleared, use the program step SF 19 before executing "**DMS**" as part of your program. Flag 19 is cleared by the subroutine if it is set.

This one contains an END, so it can go above any of the programs . . . why not just stick it on top of "**YN**"?

```
00 { 79-Byte Prgm }
01▶LBL "DMS"
02 FS?C 19
03 CLA
04 ENTER
05 STO 19
06 IP
07 CF 29
08 FIX 00
09 ARCL ST X
10 ├"°"
11 -
12 100
13 X
14 ABS
15 STO 18
16 IP
17 XEQ 01
18 ARCL ST X
19 ├"'"
20 RCL 18
21 FP
22 100
23 X
24 FIX 02
25 RND
26 FIX 01
27 XEQ 01
28 ARCL ST X
29 ├"""
30 CLX
31 FIX 04
32 SF 29
33 RCL 19
34 RTN
35▶LBL 01
36 10
37 X<>Y
38 X<Y?
39 ├"0"
40 END
```

You can go right to the top of "**DMS**" to input this next one. Scroll up to 00 and begin input. The END will separate them when you've finished.

```
00 { 133-Byte Prgm }
01▶LBL "TO"
02 CF 02
03 SF IND 25
04 "FILE: "
05 ARCL 26
06 ARCL 27
07 XEQ 00
08 ├"␊"
09 ├"DATE:"
10 XEQ 01
11 XEQ 02
12 AVIEW
13▶LBL "TOO"
14 FS? 01
15 4
16 FS? 02
17 3
18 FS? 03
19 7
20 FS? 04
21 3
22 FS? 05
23 7
24 FS? 06
25 6
26 FS? 07
27 10
28 STO 01
29 RTN
30▶LBL 00
31 ALENG
32 6
33 -
34▶LBL 02
35 X=0?
36 ├" Not Given"
37 RTN
38▶LBL 01
39 RCL 28
40 X=0?
41 RTN
42 ├"  "
43 AIP
44 ├"/"
45 FP
46 100
47 X
48 AIP
49 ├"/"
50 FP
51 100
52 X
53 AIP
54 END
```

On page 8 you input the program, "**STA**", and we said we'd be adding some more to it. Now is the time, and here are four short programs that go onto the bottom of "**STA**".

```
23▶LBL "REP"
24 CLA
25 ARCL "A"
26 ARCL "B"
27 ARCL "C"
28 RTN
29▶LBL "TIN"
30 ASTO "A"
31 ASTO IND 24
32 ASHF
33 DSE 24
34 STO ST X
35 ASTO "B"
36 ASTO IND 24
37 ASHF
38 DSE 24
39 STO ST X
40 ASTO "C"
41 ASTO IND 24
42 DSE 24
43 STO ST X
44 CLA
45 RTN
46▶LBL "MAN"
47 SF 92
48 CLA
49 AON
50 STOP
51 AOFF
52 CF 22
53 XEQ "STOA"
54 RTN
55▶LBL "STOA"
56 ASTO "A"
57 ASHF
58 ASTO "B"
59 ASHF
60 ASTO "C"
61 ASHF
62 END
```

Go to "STA" and enter **program** mode. Scroll up to put the pointer at step 22 RTN and begin typing in the programs shown to the left.

"**REP**" is the subroutine that lets you repeat the last label if the descriptor of the current shot is the same as the last one. This can be handy for taking a series of shots like "top of bank", because the whole descriptor is input with just the one keystroke, [illegible].

"**TIN**" is short for Topo **IN** . . it's the routine that stores the descriptor into the proper registers to allow retreival under the proper shot number. It also stores a copy into variables A, B and C to allow the use of the repeat routine.

MANual is the subroutine for manual input of a descriptor, or the 'prefix' of a descriptor (the rest of the descriptor is added by menu kestroke). The 'automatic' menu can be as simple or as elaborate as you want to make it, and is described in detail later in this book.

The next group of subroutines handles the curves, when you are working with Station/Offset. Shots taken within the curve area are reduced to **radial** shots.

The subroutines also adjust the instrument or backsight information when either or both of them are within the curved portion of the alignment. The shots are not just output as station/offset along a straight baseline, but actually can be plotted relative to a 'real' alignment that includes a circular (horizontal) curve.

These subroutines can also go right on top of "**YN**".

```
00 { 302-Byte Prgm }
01▸LBL "CU"
02 STO "O"
03 R↓
04 STO "S"
05 FS? 96
06 GTO 01
07 SF 95
08 "B.C. Station?␊"
09 PROMPT
10 STO 09
11 "Delta?␊"
12 PROMPT
13 X<0?
14 SF 05
15 ABS
16 →HR
17 STO 12
18 →RAD
19 "Radius?␊"
20 PROMPT
21 STO 10
22 X
23 RCL+ 09
24 STO 23
25 RCL 12
26 →HMS
27 FS? 05
28 +/-
29 CF 21
30 "Delta = "
31 XEQ "DMS"
32 ⊢"␊Radius = "
33 ARCL 10
34 ⊢"␊"
35 AVIEW
36 SF 21
37 RCL 12
38 2
39 ÷
40 TAN
41 RCLX 10
42 ENTER
43 RCL+ 09
44 X<>Y
45 ENTER
46 ENTER
47 RCL 12
48 SIN
49 X
50 FS? 05
51 +/-
52 STO 06
53 R↓
54 RCL 12
55 COS
56 X
57 +
58 STO 05
59 SF 96
60 GTO 01
61▸LBL "CU1"
62 X<>Y
63 RCL 23
64 X<Y?
65 GTO "CU3"
66 R↓
67 RCL 09
68 X<Y?
69 GTO "CU2"
70 R↓
71 X<>Y
72 GTO "CUR"
73▸LBL "CU2"
74 -
75 X<>Y
76 FC? 05
77 +/-
78 RCL+ 10
79 →POL
80 +/-
81 RCL+ 10
82 X<>Y
83 →RAD
84 RCLX 10
85 RCL+ 09
86 X<>Y
87 GTO "CUR"
88▸LBL "CU3"
89 R↓
90 RCL- 05
91 STO 30
92 R↓
93 RCL- 06
94 STO 31
95 RCL÷ 30
96 ATAN
97 FS? 05
98 +/-
99 STO 32
100 COS
101 RCL 30
102 X<>Y
103 ÷
104 STO 33
105 RCL 32
106 RCL- 12
107 STO 34
108 SIN
109 X
110 FS? 05
111 +/-
112 RCL 33
113 RCL 34
114 COS
115 X
116 RCL+ 23
117 X<>Y
118 GTO "CUR"
119▸LBL 01
120 RCL "S"
121 RCL- 09
122 RCL÷ 10
123 →DEG
124 STO 19
125 RCL 10
126 RCL- "O"
127 STO 18
128 X<>Y
129 SIN
130 X
131 RCL+ 09
132 RCL 10
133 RCL 18
134 RCL 19
135 COS
136 X
137 -
138 END
```

After you've proof-read that set of routines, let's start with input of the main program, "**TOPO**". First, go to "**TOUT**", enter **program** mode, and then scroll upward to put the pointer at step 127, SF 20. Insert the step, LBL "TOPO", scroll up again, and add an END.

The new program now has the permanent end, and you can begin to type in the rest of the program. Take your time with the input, and proof-read sections of it as you go.

The portion of the program shown on pages 12 and 13 is not the whole program, but, for now, do this much of it.

NOTE: Steps 61 and 70 refer to a global label with the name "   ". This label has **3 blank spaces** for a name, and will be input as step 129.

```
01▶LBL "TOPO"
02 XEQ "CL"
03 XEQ "FC0"
04 SF 82
05 XEQ "FILE"
06 XEQ "TY"
07 XEQ "TOO"
08▶LBL 03
09 35
10 STO 02
11 CF 21
12 CLMENU
13 "Height of Inst?"
14 ├"␊"
15 PROMPT
16 STO 11
17 "1st Shot #?␊"
18 PROMPT
19 SF 21
20 CLMENU
21 STO 13
22 RCLX 01
23 RCL+ 02
24 STO 24
25 SF 09
26▶LBL 04
27 "Inst. @?␊"
28 FC? 20
29 ├" N ↑ E"
30 FS? 20
31 ├"Sta. ↑ O/S"
32 PROMPT
33 FS? 20
34 XEQ 05
35 STO 08
36 R↓
37 STO 07
38 "Backsight?␊"
39 FC? 20
40 ├" N ↑ E"
41 FS? 20
42 ├"Sta. ↑ O/S"
43 PROMPT
44 FS? 20
45 XEQ 05
46 STO 04
47 R↓
48 STO 03
49 RCL- 07
50 RCL 04
51 RCL- 08
52 X<>Y
53 →POL
54 R↓
55 360
56 X<>Y
57 X<0?
58 +
59 STO 00
60 FS? 95
61 GTO "   "
62 CF 21
63 "Curve in Topo "
64 ├"Area?"
65 AVIEW
66 SF 21
67 XEQ "YN"
68 FS? 10
69 XEQ "CU"
70 GTO "   "
71 RTN
72▶LBL 05
73 CF 21
74 "ON CURVE?"
75 AVIEW
76 SF 21
77 XEQ "YN"
78 FS? 10
79 XEQ "CU"
80 RTN
81▶LBL 50
82 XEQ 07
83 "LABEL SHOT #"
84 XEQ 09
85 "LABEL"
86 KEY 1 GTO 01
87 "REPT"
88 KEY 3 GTO 03
89 "ADD"
90 KEY 4 GTO 04
91 "O/S"
92 KEY 5 GTO 05
93 "·MAN"
94 KEY 6 GTO 06
95 MENU
96 STOP
97▶LBL 01
98 GTO E
99▶LBL 03
100 XEQ "REP"
101 XEQ "TIN"
102 GTO "   "
103▶LBL 04
104 STO 29
105 SF 90
106 RTN
107▶LBL 05
108 STO 29
109 SF 91
110 RTN
111▶LBL 07
112 CF 29
113 FIX 00
114 CF 21
115 RTN
116▶LBL 09
117 ARCL 13
118 AVIEW
119 SF 21
120 SF 29
121 FIX 04
122 CLA
123 CLMENU
124 RTN
125▶LBL 06
126 XEQ "MAN"
127 CF 22
128 GTO E
129▶LBL "   "
130 FC?C 09
131 ISG 13
132 STO ST X
133 RCL 13
134 RCLX 01
135 RCL+ 02
136 STO 24
137 XEQ 07
138 "INPUT SHOT #"
139 XEQ 09
140 " H < "
141 KEY 1 GTO 01
142 " Z < "
143 KEY 2 GTO 02
144 "S.D."
145 KEY 3 GTO 03
146 "ROD"
147 KEY 4 GTO 04
148 MENU
149 STOP
150▶LBL 01
151 FS? 83
152 XEQ 08
153 →HR
154 RCL+ 00
155 STO 16
156 CLX
157 STOP
158▶LBL 02
159 FS? 83
160 XEQ 08
161 →HR
162 STO 17
163 STOP
164▶LBL 03
165 STO 14
166 FS? 83
167 XEQ 08
168 FIX 02
169 RCL 15
170 CF 21
171 " Rod = "
172 ARCL ST X
173 ├"↑"
174 AVIEW
175 SF 21
176 STOP
177▶LBL 04
178 STO 15
179 FS? 83
180 XEQ 08
```

Continue with the input of the additional steps below, and we'll take a breather to look at what we have so far.

The first part of the program uses subroutines to clear 'old' data and flag settings, then calls the subroutines "**FILE**", "**TY**" and "**TOO**". "**FILE**" prompts for data that will later label your output with the file name and date.

"**TY**" calls up the menus for selection (see page 7) of type of data to be stored. "**TOO**" sets the proper flags (based on your responses to the prompts so far) to have the input/output configuration stored as part of the file.

The prompts for setup information come next, the first of these being `Height of Inst?` . This prompt wants the actual **elevation** at the center of the scope, not the 'plus rod' from the point you are over. The prompt, `Inst. @?`, expects input of 2 pieces of data (either northing, ENTER, easting or station ENTER, offset). The same applies to the prompt for the backsight information.

If you are working in station/offset form, the additional prompt to determine if the points are on a curve will also appear, requiring a **yes** or **no** answer. If either or both of the points are on a curve, prompts for curve data input will appear.

If neither the instrument nor backsight station were on a curve a prompt to determine whether or not there is a curve in the topo area will appear.

```
▶LBL D
RCL 17
COS
ACOS
ENTER
SIN
RCLX 14
FS?C 90
RCL+ 29
STO 20
R↓
COS
RCLX 14
RCL+ 11
RCL- 15
STO 22
RCL 16
RCL 20
FS?C 91
XEQ 05
→REC
RCL+ 07
X<>Y
RCL+ 08
FS? 95
GTO "CU1"
▶LBL "CUR"
RCL 22
R↓
R↓
STO IND 24
DSE 24
STO ST X
R↑
STO IND 24
DSE 24
STO ST X
R↑
STO IND 24
DSE 24
STO ST X
SF 00
GTO 50
▶LBL 05
RCL 29
X<>Y
÷
ATAN
+
RCL 20
RTN
▶LBL 08
STO IND 24
DSE 24
STO ST X
RTN
```

This allows for the curved portion to be included in the calculations even if the instrument and backsight were both on a tangent.

the menu system

After the input of the rod height, the first of the menus appears, with the prompt to label the shot:

LABEL | | REPT | ADD | O/S | αMAN

This is as good a place as any to stop and discuss how the menus can be used to best advantage.

ADD is a key that allows you to 'add' or subtract from a shot in those cases where the shot could not be taken directly on the item being shot.

If, for instance, you take a shot to a 14" diameter power pole with the rod held at the face of the pole, you can input .6 and stroke this key to make the shot equivalent to a shot at the **center** of the pole.

O/S lets you handle a problem quite often encountered when the shot is to a tree. You won't be able to see the rod through the branches, so the rodperson holds it to one side . . . let's say 6' to the left. Input [6] [+/-] and use this key to adjust the shot to the **center** of the tree.

αMAN can be used to manually label a shot, or to input a prefix. The calculator is automatically put into alpha mode for input, a descriptor may be typed in, and stroking [R/S] will bring up a menu TREE BLDG STRC FENC WALL ROAD.

If the descriptor was complete, stroke [R/S] again, to complete the shot. If not completely labeled, the portion input so far will be a prefix to the rest of the descriptor.

LABEL brings up the same menu as above, without first going through αMAN. The first two, TREE and BLDG, add the word to the existing descriptor to complete it.

The remainder of the menu selections will bring up additional menu choices, containing **types**, then add to the descriptor after the next selection.

REPT is a timesaver if the description of the current shot is the same as the last shot. Just stroke this key and the label from the last shot is copied onto the current shot.

When you are adding the descriptor to your shot you can use the keys in combination. For instance, you can use [illegible] to bring up the **alpha mode** and type in 6', stroke R/S [illegible]. [illegible] brings up a new menu, with options*

BARB WW BOARD PICKT STON GATE

for selection of the next part of the descriptor.

Let's say that you select "BOARD" . . . when the shots have been downloaded later, the descriptor will be printed out as "6' board fence", and the output shot will look like the one to the right.

```
Shot #4

Sta. 11+74.533
O/S = 3.00 Right
 EL= 99.22

6' Board Fence
```

If you think of the menu key that just adds a word (TREE, BLDG) as **type A**, and the menu key (FENC, WALL) which adds a description and **then** adds a word, as being **type B**, you can see how the system can not only be customized, but also expanded, to suit your needs.

The **type B** routines add a suffix after the descriptive. There is also a **type C** (STRET, ROAD), which does not add the actual word, "street" or "road" to the descriptor, but brings up a new menu for selection.

In the programming that follows we have used "Centerline", "Top of Curb" and "Flowline" as selections. There is also an "Edge of" (LBL 30) which appends either "Pavement" or "Dirt Road", depending on the settings of flags 93 and 94.

If you study the program steps on page 16, following the paths of one of each type, you will see how to design your own descriptor system. The first descriptor menu is LBL E, and we have begun a second menu (LBL e), by assigning keys 7 and 8 to **GTO e**.

This menu only has one entry, UTLY, as an example. You can add any others which may occur to you as you use the program.

* The user may vary the menu items to suit his/her own needs. The menus shown in the text are as currently programmed on page 16.

```
▶LBL E
FC? 92
CLA
CLMENU
"TREE"
KEY 1 GTO 21
"BLDG"
KEY 2 GTO 22
"STRET"
KEY 3 GTO 23
"FENC"
KEY 4 GTO 24
"WALL"
KEY 5 GTO 25
"ROAD"
KEY 6 GTO 26
KEY 7 GTO e
KEY 8 GTO e
FC? 92
CLA
MENU
STOP
FC? 22
XEQ "REP"
FC? 22
GTO "   "
▶LBL e
CLMENU
"UTLY"
KEY 1 GTO 31
KEY 7 GTO E
KEY 8 GTO E
MENU
STOP
RTN
▶LBL 21
FS?C 92
XEQ 02
├" TREE "
GTO 01
▶LBL 22
FS?C 92
XEQ 02
├" BLDG"
GTO 01
▶LBL 23
CLA
CLMENU
"CTR"
KEY 1 GTO 27
"T.C."
KEY 2 GTO 28
"F.L."
KEY 3 GTO 29
"E.P."
KEY 4 GTO 30
FS? 93
XEQ 03
MENU
STOP
▶LBL 27
CLA
"Centerline"
GTO 01
▶LBL 28
CLA
"Top of Curb"
GTO 01
▶LBL 29
CLA
"Flowline"
GTO 01
▶LBL 30
CLA
"Edge of "
FC? 93
├"Pavement"
FS?C 94
├"Dirt "
FS?C 93
├"Road"
GTO 01
RTN
▶LBL 24
CLA
CLMENU
"BARB"
KEY 1 GTO 34
"WW"
KEY 2 GTO 35
"BOARD"
KEY 3 GTO 36
"PICKT"
KEY 4 GTO 37
"STON"
KEY 5 GTO 38
"GATE"
KEY 6 GTO 39
CLA
FS?C 92
XEQ 02
MENU
STOP
▶LBL 34
├"Barb Wire"
GTO 04
▶LBL 35
├" W. W."
GTO 04
▶LBL 36
├" Board"
GTO 04
▶LBL 37
├" Picket"
GTO 04
▶LBL 38
├" Stone"
GTO 04
▶LBL 39
├" GATE"
GTO 01
▶LBL 04
├" Fence"
GTO 01
▶LBL 25
CLA
CLMENU
"FRAME"
KEY 1 GTO 41
"STUCO"
KEY 2 GTO 42
"CMU"
KEY 3 GTO 43
"CONC"
KEY 4 GTO 44
"ROCK"
KEY 5 GTO 45
"GATE"
KEY 6 GTO 39
CLA
FS?C 92
XEQ 02
MENU
STOP
▶LBL 41
├" Frame"
GTO 05
▶LBL 42
├" Stucco"
GTO 05
▶LBL 43
├" CMU"
GTO 05
▶LBL 44
├" Conc"
GTO 05
▶LBL 45
├" Rock"
▶LBL 05
├" Wall"
GTO 01
▶LBL 26
SF 93
GTO 23
▶LBL 03
"DIRT"
KEY 5 GTO 32
"PAV'D"
KEY 6 GTO 33
RTN
▶LBL 01
XEQ "TIN"
GTO "  "
▶LBL 02
CLA
ARCL "A"
ARCL "B"
ARCL "C"
RTN
▶LBL 32
SF 94
RTN
▶LBL 33
CF 93
END
```

the labeling sequence

To better understand how the system does the writing for you, lets look at what happens in the program when we want to label a shot "6' Board Fence", a **type B** case.

First, stroke ALPHA to enter the alpha mode. Stroke 6 ▼ PUNC ▼ and then R/S. This brings up the item menu in Label E.

```
LBL E
FC? 92
CLA
CLMENU
"TREE"
KEY 1 GTO 21
"BLDG"
KEY 2 GTO 22
"STRET"
KEY 3 GTO 23
"FENC"
KEY 4 GTO 24
"WALL"
KEY 5 GTO 25
"ROAD"
KEY 6 GTO 26
KEY 7 GTO e
KEY 8 GTO e
FC? 92
CLA
MENU
STOP
```

TREE BLDG STRE FENC WALL ROAD

Each of the keys is assigned to a specific label, so when you stroke FENC, key 4 acts as the branch, and sends you to Label 24. Since this is a type B, Label 24 brings up another menu.

```
LBL 24
CLA
CLMENU
"BARB"
KEY 1 GTO 34
"WW"
KEY 2 GTO 35
"BOARD"
KEY 3 GTO 36
"PICKT"
KEY 4 GTO 37
"STON"
KEY 5 GTO 38
"GATE"
KEY 6 GTO 39
CLA
FS?C 92
XEQ 02
MENU
STOP
```

This menu acts as a prompt for 'what kind' of fence. When you select Board, key 3 sends you to Label 36, which appends the word "Board" to what is already in the alpha register, and then sends you to Label 04.

```
LBL 36
⊢" Board"
GTO 04
```

Label 04 appends the word "Fence" to the alpha register, then sends you to Label 01 to finish the process of storing the descriptor.

```
LBL 04
⊢" Fence"
GTO 01
```

```
LBL 01
XEQ "TIN"
GTO "
```

If you will look again at Label E on page 17, notice that steps 253 and 254 assign **Label e** to keys 7 and 8. These are the 'scroll' keys, ▼ and ▲.

```
263▶LBL e
264 CLMENU
265 "UTLY"
266 KEY 1 GTO 31
267 KEY 7 GTO E
268 KEY 8 GTO E
269 MENU
270 STOP
271 RTN
```

Lable e, in turn assigns these two keys back to Label E. This is how you set up a second menu. In this case we only have one key assigned, and it goes to UTLY, for utilities, in Label 31.

Additional items may be added to the menu, or UTLY can call up an additional menu of just utility items. As an example, the short program below may be substituted for the original Label e, to handle a lot of the work.

The first menu types in a prefix, the second a suffix, allowing you to spell out "Tele. Pole", "Water Valve", "Storm Manhole", etc.

```
▶LBL e
 CF 22
 FC? 92
 CLA
 CLMENU
 "ELEC"
 KEY 1 GTO 51
 "TELE"
 KEY 2 GTO 52
 "CATV"
 KEY 3 GTO 53
 "STOM"
 KEY 4 GTO 54
 "SAN"
 KEY 5 GTO 55
 "H2 O"
 KEY 6 GTO 56
 KEY 7 GTO 57
 KEY 8 GTO 57
 FC? 92
 CLA
 MENU
 STOP
 FC? 22
 GTO 58
▶LBL 51
 ├" Elec."
 GTO 57
▶LBL 52
 ├" Tele."
 GTO 57
▶LBL 53
 ├" CATV"
 GTO 57
▶LBL 54
 ├" Storm"
 GTO 57
▶LBL 55
 ├" San."
 GTO 57
▶LBL 56
 ├" Water"
▶LBL 57
 XEQ "STOA"
 CLA
 CF 22
 CLMENU
 "POLE"
 KEY 1 GTO 61
 "MH"
 KEY 2 GTO 62
 "BOX"
 KEY 3 GTO 63
 "VALVE"
 KEY 4 GTO 64
 "INLET"
 KEY 5 GTO 65
 "HYD"
 KEY 6 GTO 66
 MENU
 XEQ "REP"
 STOP
 FC? 22
 GTO 58
 GTO "AA"
▶LBL 61
 ├" Pole"
 GTO 58
▶LBL 62
 ├" Manhole"
 GTO 58
▶LBL 63
 ├" Box"
 GTO 58
▶LBL 64
 ├" Valve"
 GTO 58
▶LBL 65
 ├" Inlet"
 GTO 58
▶LBL 66
 ├" Hydrant"
▶LBL 58
 XEQ "TIN"
 GTO "  "
```

With these examples, you should be able to write menus that best suit your own needs.

the output

There is an illustration, on page 15, that shows the output of a typical shot. That example is of a station/offset topo which stored the finished data and the descriptor.

```
Shot #4
H<= 9°05'25.0"
Z<= 90°03'00.0"
S.D.= 101.28
ROD= 0.00

 N = 4,398.7590
 E = 4,997.1893
 EL= 92.66

16" OAK TREE
```

The example to the left is of a shot in coordinate topo, and in this case the raw data was also stored. Storing the raw data, finished data and descriptor requires 10 registers, while just storing the finished data and descriptor only requires 6.

The number of shots that may be stored is, of course, directly proportional to the amount of memory you have left after you have the program and menus in their final form. You can check available memory by sizing the calculator to **0035**, and then checking the amount of bytes left (MEM, through CATALOG). Dividing this number by 9, and then adding 35 to it should tell you about what size will work, in terms of registers.

You can again check the number of bytes through MEM, and if you forget what size you are set to, RCL REGS will bring up a matrix as an answer. The first number in the matrix is your current size.

more on prompts

Most of the prompts were described as we wrote the subroutines which call them up, but here are a few additional comments:

1. The prompts for file name and date may be answered with R/S if you don't need them. The output will show "not given" when the file is output.

2. The date should be input as month, decimal point, day, last two digits of the year.

3. If you are working in **coordinate mode**, answer the prompt, "curve in topo area?" NO.

4. If you are working in station/offset mode, a curve to the left is input with a negative delta. Stroke +/- before R/S if the curve is to the left.

5. Data **must** be output before additional use of the calculator, to protect the calculator's "environment". There are a number of flag settings and code numbers in storage that tell the output program what to do. Disturbing these settings will prevent output (most likely, "alpha data invalid" when a wrong register is recalled).

debugging assistance

We have always tried to give as much assistance as possible to our users by telephone, but with this program it is not too practical. If you experience difficulty in getting the program to work, send us a note explaining what the program is doing wrong, a print-out of the program and subroutines, and a self-addressed, stamped envelope.

We will try to proof-read the program for you and make suggestions by return mail.

Software by D'Zign
P.O. Box 430 • Tollhouse, CA 93667

The Most Commonly Asked Questions

The following questions and answers were compiled from the calls and letters we've received in the past 4+ years that we've been publishing solution books for the HP42S calculator, and are included here in the event that your question is one of them.

Q: *How do you type in the* ***END****?*
A: There are a number of ways . . . one easy way is to stroke XEQ ENTER and type it in, using the alpha keys. Because you stroked XEQ first, the calculator will recognize that this is not an alpha input, and substitute the actual function when you stroke ENTER again. You may input *any* function by this method.

You may also take advantage of the built-in *function catalog*, stroke ▢ + (catalog), and then the FCN menu key. You may scroll up or down with the ▲ or ▼ keys, and ***all*** of the calculator's functions are in there. When you reach the one you want, just stroke the key under the menu item.

Q: *How do you type in the* ***indirect*** *calls, such as step 12 in the first program on page 5?*
A: The indirect calls are made by stroking •. In the case of the call above, first stroke ▢ 6 (flags), then CF, to bring up the prompt CF__ , then stroke •. Some of the indirect calls give a secondary prompt, requiring another •.

Q: *How do I type in a* ***ARCL*** *command?*
A: Enter ***alpha mode*** before stroking RCL or STO.

D'Zign
P.O. Box 430
Tollhouse, CA 93667-0430

$10.00 U.S.

ISBN 0-944889-09-3